

# Example of SAS Weave

## Documenting your programs

# Documenting code and Outputs

# Class Height and Weight by Sex

## Two different Y axes

```
ods pdf bookmarkgen=off;
proc report data=sashelp.class split="/" nocenter missing ;
column ("Sashelp class data set for plot" name sex age height weight) _row;
define name / display format= $8. width=8 spacing=2 left "Name" ;
define sex / display format= $1. width=1 spacing=2 center "Gender" ;
define age / sum format= best9. width=9 spacing=2 center "Age" ;
define height / sum format= best9. width=9 spacing=2 center "Height" ;
define weight / sum format= best9. width=9 spacing=2 center "Weight" ;
%greenbar;
run;quit;
```

Sashelp class data set for plot


Name	Gender	Age	Height	Weight
Alfred	M	14	69	112.5
Alice	F	13	56.5	84
Barbara	F	13	65.3	98
Carol	F	14	62.8	102.5
Henry	M	14	63.5	102.5
James	M	12	57.3	83
Jane	F	12	59.8	84.5
Janet	F	15	62.5	112.5
Jeffrey	M	13	62.5	84
John	M	12	59	99.5
Joyce	F	11	51.3	50.5
Judy	F	14	64.3	90
Louise	F	12	56.3	77
Mary	F	15	66.5	112
Philip	M	16	72	150
Robert	M	12	64.8	128
Ronald	M	15	67	133
Thomas	M	11	57.5	85
William	M	15	66.5	112

```

ods pdf bookmarkgen=off;
proc template;
  define statgraph plot;
    begingraph;
      entrytitle "Series plot with additional Y2AXIS" ;
      layout datapanel classvars=(sex) / columns=2 rows=1
        headerlabelattrs=(color=red weight=bold)
        headeropaque=false;
      layout prototype;
        seriesplot x=age y=height / group=dose display=all name="height"
          lineattrs=(color=cx5DAF5D)
          markerattrs=(symbol=squarefilled color=cx5DAF5D);
        seriesplot x=age y=weight / group=dose
          yaxis=y2 display=all name="weight"
          lineattrs=(color=cx4B50AA pattern=2)
          markerattrs=(symbol=circlefilled color=cx4B50AA);
      endlayout;
      sidebar / align=bottom;
        discretelegend "height" "weight" / ;
      endsidebar;
    endlayout;
  endgraph;
end;
define style noheaderborder;
  parent = styles.default;
  class graphborderlines / contrastcolor=white;
  class graphbackground / color=white ;
end;
proc sort data=sashelp.class out=class;
by age;
run;quit;
proc sgrender data=class template=plot;
run;quit;

```

Series plot with addtional Y2AXIS



Class Height and Weight by Sex  
Four plots on one page



```
ods pdf bookmarkgen=off;
proc template;
  define statgraph panel;
    begingraph;
      entrytitle "paneled display ";
      layout lattice / rows = 2 columns = 2 rowgutter = 10
        columngutter = 10;
      layout overlay; scatterplot y = weight x = height;
        regressionplot y = weight x = height;
      endlayout;
      layout overlay / xaxisopts = (label = "weight");
        histogram weight;
      endlayout;
      layout overlay / yaxisopts = (label = "height");
        boxplot y = height;
      endlayout;
      layout overlay; scatterplot y = weight x = height /
        group = sex name = "scat";
        discretelegend "scat" /
          location=inside autoalign=(topleft) across=1;
      endlayout;
    endlayout;
  endgraph;
end;
run;quit;
proc sgrender data = sashelp.class template = panel;
run;quit;
```

paneled display

